
1 [bookmark: _Toc122434485][bookmark: _Toc165562569]Overview
This document lists all the changes needed to correct issues in the ATS iwd-TTCN3-B2022-09_D24wk12 related to the title of this CR.
Contact:	Olivier Genoud
	olivier.genoud@etsi.org
2 [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc165562570]Corrections required
Change 1
	Function names
	f_MCX_SIP_RegistrationSubscription_DecodeMatchAndCheckMikeyMessage_SetCSK, a_MCX_SIP_ServiceAuthorization_ReceiveSubscribeSend200OK

	Reason for change
	R5-240565 has corrected the MIKEY message for CSK upload for the case of the SIP REGISTER but breaking the case of the CSK upload to CMS and GMS with SIP SUBCRIBE

	Summary of change
	Parameter p_RequestUri := omit added to f_MCX_SIP_RegistrationSubscription_DecodeMatchAndCheckMikeyMessage_SetCSK to allow a_MCX_SIP_ServiceAuthorization_ReceiveSubscribeSend200OK to hand over the correct URI in case of CSK upload to GMS/CMS

	TTCN module
	MCX_SIP_Registration



	  function f_MCX_SIP_RegistrationSubscription_DecodeMatchAndCheckMikeyMessage_SetCSK(MCX_ServiceType_Type p_ServiceType,
                                                                                     MIME_Message p_MIME_Message,
                                                                                     MCX_KeyIndex_Type p_KeyIndex,
                                                                                     template (omit) SipUrl p_RequestUri := omit) runs on MCX_IMS_PTC
  {
    var charstring v_IDRr;
    var template (present) MIKEY_Message v_MIKEY_MessageTemplate;
    var MIKEY_Message v_MIKEY_Message;

    if (isvalue(p_RequestUri)) {
      v_IDRr := f_SIP_SipUrl2Str(valueof(p_RequestUri));
    } else {
      v_IDRr := f_MCX_MCServer_PublicServiceId(p_ServiceType);
    }
    v_IDRr := f_MCX_MCServer_PublicServiceId(p_ServiceType);
    v_MIKEY_MessageTemplate := f_MIKEY_SAKKE_MessageRX_CSK(p_ServiceType, v_IDRr);
    v_MIKEY_Message := f_MCX_MIME_DecodeMatchAndCheckMikeyMessage(p_MIME_Message, v_MIKEY_MessageTemplate);
    fl_MCX_IMS_MIKEY_Message_ExtractAndSetCSK(v_MIKEY_Message, p_KeyIndex);   // p_KeyIndex may be CSK, CSK_CMS, CSK_GMS1, CSK_GMS2
  }

	  altstep a_MCX_SIP_ServiceAuthorization_ReceiveSubscribeSend200OK(out MCX_DataRequestWithXmlBodies_Type p_DataRequestWithXmlBodies,
                                                                   MCX_ServiceType_Type p_ServiceType,
                                                                   MCX_Subscription_Type p_Subscription := INITIAL_SUBSCRIBE,
                                                                   IMS_PTC_DialogIndex_Type p_DialogIndex,
                                                                   SipUrl p_RequestURI,
                                                                   SipUrl p_ContactURI,
                                                                   template (present) ResourceLists_Type p_ResourceListsTemplate) runs on MCX_IMS_PTC
  {
    var MCX_DataRequestWithXmlBodies_Type v_DataRequestWithXmlBodies;
    var IMS_DATA_REQ v_IMS_SUBSCRIBE_REQ_ByRef;
    var SUBSCRIBE_Request v_SubscribeRequest;
    var MIME_Message v_MIME_Message;
    var MCX_KeyIndex_Type v_KeyIndex;
    var template (present) MCX_ServiceInfo_Type v_MCS_Info := cs_MCX_Service_Info_Init;
   
    [] a_MCX_SIP_ReceiveSubscribeSend200OK(v_IMS_SUBSCRIBE_REQ_ByRef, p_ServiceType, p_Subscription, p_DialogIndex, p_RequestURI, p_ContactURI) {
      v_DataRequestWithXmlBodies.Request := v_IMS_SUBSCRIBE_REQ_ByRef.Request;
      v_SubscribeRequest := v_DataRequestWithXmlBodies.Request.Subscribe;
      v_MIME_Message := v_SubscribeRequest.messageBody.mimeMessageBody;
      v_KeyIndex := fl_MCX_XcapDiffDialog_GetKeyIndex(p_DialogIndex);

      if (p_Subscription == INITIAL_SUBSCRIBE) {
        // Check MIKEY message according to 36.579-1 Table 5.5.9.1-1: MIKEY message from the UE
        f_MCX_SIP_RegistrationSubscription_DecodeMatchAndCheckMikeyMessage_SetCSK(p_ServiceType, v_MIME_Message, v_KeyIndex, v_SubscribeRequest.requestLine.requestUri);
        v_MCS_Info := fl_MCX_SIP_Registration_McsInfo(p_ServiceType, v_DataRequestWithXmlBodies.Request);
      }
      
      v_DataRequestWithXmlBodies.XmlBodies := f_MCX_MIME_DecodeAndMatchXML(v_MIME_Message, p_ServiceType, v_KeyIndex, v_MCS_Info.Mcptt, -, v_MCS_Info.Mcvideo, -, v_MCS_Info.Mcdata, -, -, -, -, p_ResourceListsTemplate);

      // Store client-id (if any: e.g. there is no MCPTT info in re-SUBSCRIBE)
      fl_MCX_SIP_Registration_ClientId_Set(v_DataRequestWithXmlBodies.XmlBodies);

      p_DataRequestWithXmlBodies := v_DataRequestWithXmlBodies;
    }
  }



Change 2
	Function names
	f_MCX_SecurityParameters_Init

	Reason for change
	According to 33.180 F.2.1 the Current Key Period No (P5) is calculated by 
		P5 = Floor ( ( TIME - P4 ) / P3 )
but in f_MCX_SecurityParameters_Init the Key Period offset (P4) is missing in the calculation

	Summary of change
	Calculation of the Current Key Period No corrected

	TTCN module
	MCX_SecurityParameters



	  function f_MCX_SecurityParameters_Init(charstring p_KSM_Id) return MCX_SecurityParameters_Type
  {
    var MCX_SecurityParameters_Type v_SecurityParameters := valueof(cs_MCX_SecurityParameters_Init);
    var integer v_TimestampSeconds := f_UTC_TimestampSeconds();
    var MCX_SecurityKey_Type v_GMK := f_MCX_SecurityParameters_CreateKey(GMK);
    var MCX_SecurityKey_Type v_PCK := f_MCX_SecurityParameters_CreateKey(PCK);

    v_SecurityParameters := f_MCX_SecurityParameters_SetKey(v_SecurityParameters, GMK, v_GMK);
    v_SecurityParameters := f_MCX_SecurityParameters_SetKey(v_SecurityParameters, PCK, v_PCK);
    v_SecurityParameters := f_MCX_SecurityParameters_SetKey(v_SecurityParameters, CSK);            // initialised as omit
    v_SecurityParameters := f_MCX_SecurityParameters_SetKey(v_SecurityParameters, CSK_CMS);        // initialised as omit
    v_SecurityParameters := f_MCX_SecurityParameters_SetKey(v_SecurityParameters, CSK_GMS1);       // initialised as omit
    v_SecurityParameters := f_MCX_SecurityParameters_SetKey(v_SecurityParameters, CSK_GMS2);       // initialised as omit
    v_SecurityParameters := f_MCX_SecurityParameters_SetKey(v_SecurityParameters, CSK_GMS3);       // initialised as omit @sic R5s210398 sic@
    v_SecurityParameters := f_MCX_SecurityParameters_SetKey(v_SecurityParameters, CSK_GMS4);       // initialised as omit @sic R5s210398 sic@
    v_SecurityParameters := f_MCX_SecurityParameters_SetKey(v_SecurityParameters, TrK, cs_MCX_SecurityKey(px_MCX_TrK_ID, bit2oct(px_MCX_TrK)));
    v_SecurityParameters := f_MCX_SecurityParameters_SetKey(v_SecurityParameters, InK, cs_MCX_SecurityKey(px_MCX_InK_ID, bit2oct(px_MCX_InK)));

    v_SecurityParameters.KSM_Id := p_KSM_Id;
    v_SecurityParameters.ECCSI.KPAK := fx_ECCSI_GenerateKPAK(v_SecurityParameters.ECCSI.KSAK);
    v_SecurityParameters.SAKKE.PublicKey := fx_SAKKE_GeneratePublicKey(v_SecurityParameters.SAKKE.MasterSecret);
    v_SecurityParameters.KeyPeriod.Offset := v_TimestampSeconds mod v_SecurityParameters.KeyPeriod.Length;
    v_SecurityParameters.KeyPeriod.No := (v_TimestampSeconds - v_SecurityParameters.KeyPeriod.Offset) / v_SecurityParameters.KeyPeriod.Length;

    return v_SecurityParameters;
  }



