
1 [bookmark: _Toc163647039]Overview
[bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc162972719]This document lists all the essential changes needed to correct problems in the iwd-TTCN3-B2022-09_D24wk12 related to the title of this CR. 
Contact: 	Juan Garcia Martin
	Juan.GarciaMartin@rohde-schwarz.com
	
2 [bookmark: _Toc163647040]Corrections required:
2.1 [bookmark: _Toc325725665][bookmark: _Toc122434493][bookmark: _Toc295288970][bookmark: _Toc163647041]Correction to function f_NBIOT_ModifySysinfoCombinationFromC1_To_C8
	Function name
	f_NBIOT_ModifySysinfoCombinationFromC1_To_C8

	Reason for change
	Wrong System information combination type in function f_NBIOT_ModifySysinfoCombinationFromC1_To_C8.

	Summary of change
	Changed system information combination form c1 to c8.

	TTCN module
	NBIOT_CellInfo



Before Change:
	function f_NBIOT_ModifySysinfoCombinationFromC1_To_C8(NBIOT_CellId_Type p_CellId,
                                           template (omit) SystemInformationBlockType31_NB_r17.servingSatelliteInfo_r17.ul_SyncValidityDuration_r17 p_UL_SyncValidityDuration_r17 := omit,
                                           boolean p_DoAttachWithoutPDN := true,  //AllowAttachWithoutPDN = CONTROL_PLANE
                                           integer p_FrequencyBand := px_ePrimaryFrequencyBand,
                                           EUTRA_NAS_NumOfPLMN_Type p_CellsOnPLMN := LTE_AllCellsOnSamePLMN) runs on NBIOT_PTC
  {
    ……
    //Update scheduling info list
    v_SIB1.message_.c1.systemInformationBlockType1_r13.schedulingInfoList_r13 := v_SchedulingInfoList;
    v_SIB1.message_.c1.systemInformationBlockType1_r13.nonCriticalExtension := cs_SystemInformationBlockType1_NB_v1350_NTN_r17(v_TAC_CellId);
    
    v_CellBcchInfo := cs_BCCHInfo_NB_C1(v_MIB, v_SIB1, v_SI_List);
    f_NBIOT_CellInfo_SetBCCH_Info(p_CellId, v_CellBcchInfo);
    f_NBIOT_CellInfo_SetCombination(p_CellId, c1);
  }



After Change:
	function f_NBIOT_ModifySysinfoCombinationFromC1_To_C8(NBIOT_CellId_Type p_CellId,
                                           template (omit) SystemInformationBlockType31_NB_r17.servingSatelliteInfo_r17.ul_SyncValidityDuration_r17 p_UL_SyncValidityDuration_r17 := omit,
                                           boolean p_DoAttachWithoutPDN := true,  //AllowAttachWithoutPDN = CONTROL_PLANE
                                           integer p_FrequencyBand := px_ePrimaryFrequencyBand,
                                           EUTRA_NAS_NumOfPLMN_Type p_CellsOnPLMN := LTE_AllCellsOnSamePLMN) runs on NBIOT_PTC
  {
          
    //Update scheduling info list
    v_SIB1.message_.c1.systemInformationBlockType1_r13.schedulingInfoList_r13 := v_SchedulingInfoList;
    v_SIB1.message_.c1.systemInformationBlockType1_r13.nonCriticalExtension := cs_SystemInformationBlockType1_NB_v1350_NTN_r17(v_TAC_CellId);
    
    v_CellBcchInfo := cs_BCCHInfo_NB_C1(v_MIB, v_SIB1, v_SI_List);
    f_NBIOT_CellInfo_SetBCCH_Info(p_CellId, v_CellBcchInfo);
    f_NBIOT_CellInfo_SetCombination(p_CellId, c8);
   
  }



2.2 [bookmark: _Toc163647042]Correction to function f_TC_22_4_30_NBIOT
	Function name
	f_TC_22_4_30_NBIOT 

	Reason for change
	Current implementation of TTCN( t_T317_min, t_T317_max, t_T318_min), makes it difficult to ensure that at Step 26A, the SystemInformationBlockType31-NB is broadcasted to the UE when UE starts reading SIBs after T318 timer expiry at the UE side, if the timer on the SS side are not aligned to the UE timers then we could result in test case failure.
Add new timer of  t_T317_Step24B, that takes the K-Offset into account       (to align the start timer T-317 more accurately, and start after SS receive the UL Data(Step 24B) from UE, it is more accurate to sync up with timer (T317 and T318) in UE side.

	Summary of change
	Added new timer of t_T317_Step24B (which takes K_Offset into account).
Replaced the t_T317_min, t_T317_max with t_T317_Step24B.
Replaced the t_T318_min with normal t_T318.

	TTCN module
	 NBIOT_RRC_NTN



Before Change:
	function f_TC_22_4_30_NBIOT() runs on NBIOT_PTC
  {// NB-IoT / NTN / Ephemeris information update / T317 Expiry / T318 Expiry @sic R5-235429, R5s230708 Ch2.1 sic@
    var NAS_MSG_Indication_Type v_NAS_Ind;
    var O10_Type v_UserData;
    var SystemInformationBlockType2_NB_r13 v_SIB2;
    var DRB_IdentityList_Type v_DrbIdList;
    var integer v_SelectedPlmnIndex;

      
    timer t_T317 := 60.0; //Corresponds to the value (60sec) of ul_SyncValidityDuration_r17 in SIB31-NB acc. Table 22.4.30.3.3-2
    timer t_T318 := 8.0;  //Corresponds to the value (8000ms) of t318-r17 acc 36.508 Table 4.6.3-14
    timer t_T317_min := f_NBIOT_SetTimerToleranceMin(rrcTimer,60.0);
    timer t_T317_max := f_NBIOT_SetTimerToleranceMax(rrcTimer,60.0);
    timer t_T318_min := f_NBIOT_SetTimerToleranceMin(rrcTimer,8.0);
    timer t_T301 := 2.5;
    timer t_T5 := 30.0;
   

    //System information combination 8 as defined in TS 36.508 [18] clause 8.1.4.3.1.1 is used.
    f_NBIOT_Init(c8, CONTROL_PLANE);
      
    //Table 22.4.30.3.3-2, Value of T317=60s
    f_NBIOT_CellInfo_ModifySIB31_UL_SyncValidityDuration_r17(nbiot_Cell1,s60);
      
    //Table 22.4.30.3.3-4  @sic R5s230708 Ch2.1.2 R5-237383 sic@
    v_SIB2 := f_NBIOT_CellInfo_GetSIB2(nbiot_Cell1);
    v_SIB2.ue_TimersAndConstants_r13.t311_r13 := ms10000;
    v_SIB2.ue_TimersAndConstants_r13.t301_r13 := ms2500;
    v_SIB2.cp_Reestablishment_r14 := true_;
    f_NBIOT_CellInfo_SetSIB2 (nbiot_Cell1, v_SIB2);

    
    //Create and configure all cells
    f_NBIOT_CellConfig_Def(nbiot_Cell1, CONTROL_PLANE);
      
    f_UT_SwitchOnUE(UT, false);
    //The UE is in Automatic PLMN selection mode
    f_UT_AutomaticPLMN_Select_WithSwitchOnOff (UT);
      
    //The preconfigured UE location is defined in TS 36.508 [18] Clause 4.13.
    f_UT_ConfigureNTNPosition(UT);
    
    //Preamble: The UE is in State 1-NB switched off according to TS 36.508 [18].
    f_UT_PowerOffUE (UT);
      
    f_NBIOT_SetCellPower(nbiot_Cell1, tsc_ServingCellRS_EPRE);
    
    f_NBIOT_TestBody_Set(true);
    
    //@siclog "Step 1" siclog@
    //Power on UE.
    f_UT_SwitchOnUE(UT, false);
      
    //@siclog "Step 2" siclog@
    //UE acquires SystemInformationBlockType31-NB and start T317 = 60 seconds.
    
    //@siclog "Steps 3-18b1" siclog@
    //Steps 1 to 16b1 of the Attach, connected mode and test mode activated procedure described in TS 36.508 [18] subclause 8.1.5.2A.3 with TEST LOOP MODE A are performed on Ncell 1.
    // Acc to 36.508 Table 8.1.5.2A.3-1: NB-IoT UE Attach, connected mode and test mode activated procedure (state 2-NB to state 2A-NB)
    //Note: After receiving the message in step 4, SS start a timer T1 with value = T317.
    v_NAS_Ind := f_NBIOT_InitialRegistration_Step2_4(nbiot_Cell1, CONTROL_PLANE, PREAMBLE); //@sic R5230708 Ch2.1.3 sic@
    t_T317.start;    //T1 started
    f_NBIOT_InitialRegistration_Step5_14(nbiot_Cell1, CONTROL_PLANE, STATE2A_NB_TESTLOOP_ModeG, v_NAS_Ind);
         
    //@siclog "Steps 18A-18B" siclog@
    //Steps 1 to 2 of the UE Test Loopback Activated procedure described in TS 36.508 [18] subclause 8.1.5.2B.3 are performed on Ncell 1.
    f_NBIOT_CloseUE_TestLoopModeGH (nbiot_Cell1, tsc_UE_TestLoopMode_TypeG, tsc_TestLoopModeGH_UplinkMode_NAS); //@sic R5230708 Ch2.1.4 sic@
      
    //@siclog "Step 19" siclog@
    //SS stops broadcasting SystemInformationBlockType31-NB
    f_NBIOT_ModifySysinfoCombinationTo_C1(nbiot_Cell1, true, false);  //now only SIB2 will be broadcasted with cellBarred and  DottachWithoutPDN = false for CONTROL_PLANE
    f_NBIOT_SS_ConfigureSysinfo(nbiot_Cell1, false);
     
    //@siclog "Step 19A" siclog@
    //SS stops periodic grant allocation
    f_NBIOT_ULGrantTransmission(nbiot_Cell1, cs_NB_UL_GrantScheduling_Stop);
      
    //@siclog "Step 20" siclog@
    //The SS transmits one IP packet embedded in an ESM DATA TRANSPORT and DLInformationTransfer-NB. (Note 1)
    v_UserData := f_GetN_OctetsFromPRBS(0, 10);
    SRB.send(cas_NB_SRB_NasPdu_REQ(nbiot_Cell1,
                                     tsc_SRB1bis,
                                     cs_TimingInfo_Now,
                                     cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,
                                                    cs_ESM_DATA_TRANSPORT (tsc_EpsDefaultBearerId, tsc_PTI_Unassigned, cs_UserDataContainer(v_UserData))))); //@sic R5s230708 Ch.2.1.6 sic@
      
      
    //@siclog "Steps 21-22" siclog@
    //Void
      
    //@siclog "Step 23" siclog@
    //Wait until expiry of T1.
    //Note: UE starts timer T318. SS start a timer T2 with value = T318.
    t_T317.timeout;  //T1 expires
    t_T318.start;    //T2 started
     
    //@siclog "Step 24" siclog@
    //The SS starts the UL periodic grant transmission, allowing the UE to return the IP packet as received in step 20
    f_NBIOT_ULGrantTransmission(nbiot_Cell1, cs_NB_UL_GrantScheduling_Start(cs_NB_ContinuousGrant)); //@sic R5s230708 Ch.2.1.7 sic@
            
    //@siclog "Step 24A" siclog@
    //SS resumes broadcasting SystemInformationBlockType31-NB.
    f_NBIOT_ModifySysinfoCombinationFromC1_To_C8(nbiot_Cell1, s60, false);  //now SIB31-NB will be broadcasted again with T317=60s and DoAttachWithoutPDN=false for CONTROL_PLANE
    f_NBIOT_SS_ConfigureSysinfo(nbiot_Cell1, false);
    //Note: Timer T318 shall be stopped and T317 shall be started in UE side. SS starts a timer T3 with value = T317.
    t_T318.stop;     //T2 stopped
      
    //@siclog "Step 24B" siclog@
    //Check: Does the UE transmit one IP packet embedded in an ESM DATA TRANSPORT and ULInformationTransfer-NBcorresponding to grant in step 24?
    SRB.receive(car_NB_SRB_NasPdu_IND(nbiot_Cell1,
                                      tsc_SRB1bis,
                                      cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,
                                      cr_ESM_DATA_TRANSPORT(?, cr_UserDataContainer))));
    f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 24B");
    t_T317_max.start;
    t_T317_min.start;    //T3 started @sic R5s230708, R5-237383 sic@

     
    //@siclog "Step 24C" siclog@
    //SS stops broadcasting SystemInformationBlockType31-NB.
    f_NBIOT_ModifySysinfoCombinationTo_C1(nbiot_Cell1,true,false);  //now only SIB2 will be broadcasted with cellBarred and  DottachWithoutPDN = false for CONTROL_PLANE
    f_NBIOT_SS_ConfigureSysinfo(nbiot_Cell1, false);
      
    //@siclog "Step 24CA" siclog@
    //SS stops periodic grant allocation
    f_NBIOT_ULGrantTransmission(nbiot_Cell1, cs_NB_UL_GrantScheduling_Stop);
    
    //@siclog "Step 24D" siclog@
    //The SS transmits one IP packet embedded in an ESM DATA TRANSPORT and DLInformationTransfer-NB (Note 1).
    v_UserData := f_GetN_OctetsFromPRBS(10, 10);
    SRB.send(cas_NB_SRB_NasPdu_REQ(nbiot_Cell1,
                                     tsc_SRB1bis,
                                     cs_TimingInfo_Now,
                                     cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,
                                                    cs_ESM_DATA_TRANSPORT (tsc_EpsDefaultBearerId, tsc_PTI_Unassigned, cs_UserDataContainer(v_UserData))))); //@sic R5s230708 Ch.2.1.6 sic@
      
      
    //@siclog "Step 24E" siclog@
    //Wait until expiry of T3.
    //Note: UE starts timer T318. SS starts a timer T4 with value = T318
    t_T317_min.timeout;   //T3 expired  @sic R5s230708, R5-237383 sic@
    t_T318_min.start;     //T4 started
    t_T317_max.timeout;

      
    //@siclog "Step 24F" siclog@
    //SS starts periodic UL grant transmission, allowing the UE to return the IP packet as received in step 24D.
    f_NBIOT_ULGrantTransmission(nbiot_Cell1, cs_NB_UL_GrantScheduling_Start(cs_NB_ContinuousGrant)); //@sic R5s230708 Ch.2.1.7 sic@
     
      
    //@siclog "Step 25" siclog@
    //Check: Does the UE transmit one IP packet embedded in an ESM DATA TRANSPORT and ULInformationTransfer-NB corresponding to grant in step 24F?
    alt{
      [] SRB.receive(car_NB_SRB_NasPdu_IND(nbiot_Cell1,
                                      tsc_SRB1bis,
                                      cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,
                                      cr_ESM_DATA_TRANSPORT(?, cr_UserDataContainer))))
    
        {
            f_NBIOT_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 25");
        }
    
      [] t_T318_min.timeout  //T4 expires @sic R5s230708, R5-237383 sic@
        {
          //@siclog "Step 26" siclog@
          //Wait until expiry of T4.
        }
    }
     
    //@siclog "Step 26A" siclog@
    //SS resumes broadcasting SystemInformationBlockType31-NB.
    f_NBIOT_ModifySysinfoCombinationFromC1_To_C8(nbiot_Cell1, s60, false);  //now SIB31-NB will be broadcasted again with T317=60s and DoAttachWithoutPDN=false for CONTROL_PLANE
    f_NBIOT_SS_ConfigureSysinfo(nbiot_Cell1, false);
    
    //EXCEPTION: Steps 27-30 describe behaviour that depends on whether the UE supports RRC connection re-establishment.
    if(pc_RRC_re_establishment_CP_CIoT)
    {//@sic R5s230708 Ch2.1.9 R5-237383 sic@
        // Locally release
        v_DrbIdList := f_NBIOT_SS_ActiveDRBs(nbiot_Cell1, CONTROL_PLANE);
        f_NBIOT_SS_SRBs_DRBs_Release(nbiot_Cell1, CONTROL_PLANE, -, v_DrbIdList);
        f_NBIOT_SS_SRBs_DRBs_Config(nbiot_Cell1, CONTROL_PLANE, -, omit);
        f_NBIOT_SS_ConfigRachProcedureWithNoMsg4(nbiot_Cell1);

        //@siclog "Step 27" siclog@
        //Check: Does the UE transmit an RRCConnectionReestablishmentRequest-NB message on SRB0?
        f_NBIOT_RRC_RRCConnectionReestablishmentRequest_r14(nbiot_Cell1);
        f_NBIOT_PreliminaryPass(__FILE__, __LINE__, " Step 27");
        
        //@siclog "Step 28" siclog@
        //The SS does not respond for 2.5s (T301) and the SS shall handle any re-transmitted RRCConnectionReestablishmentRequest-NB message on Ncell 2
        t_T301.start;
        alt {
          [] SRB.receive(car_NB_SRB0_RrcPdu_IND(nbiot_Cell1, cr_RRCConnectionReestablishmentRequest_NB_r14(?, otherFailure)))
            {
              repeat;
            }
          [] t_T301.timeout {}
        }
       
        //@siclog "Step 29" siclog@
        //Void
        
        //@siclog "Step 30" siclog@
        //Void
        
        f_NBIOT_SS_ConfigRachProcedureMsg4(nbiot_Cell1, f_NBIOT_508_RRCConnectionSetup());
    }
    else
    {
        // Locally release the connection as UE will go to idle mode
        v_DrbIdList := f_NBIOT_SS_ActiveDRBs(nbiot_Cell1, CONTROL_PLANE);
        f_NBIOT_SS_SRBs_DRBs_Release(nbiot_Cell1, CONTROL_PLANE, -, v_DrbIdList);
        f_NBIOT_SS_SRBs_DRBs_Config(nbiot_Cell1, CONTROL_PLANE, -, omit);
    }
      
    
    //@siclog "Step 31" siclog@
    t_T5.start; //@sic R5s230708, R5-237383 sic@
    alt {//@sic R5s230708, R5-237383 sic@
      //@siclog "Step 32a1" siclog@
      [] SRB.receive(car_NB_SRB0_RrcPdu_IND(nbiot_Cell1, cr_RRCConnectionRequest_NB(?, ?)))
        {
          //@siclog "Step 32a2" siclog@
          t_T5.stop;

          // switch on UL grant assignments
          f_NBIOT_ULGrantTransmission(nbiot_Cell1, cs_NB_UL_GrantScheduling_Start(cs_NB_ContinuousGrant));
          f_NBIOT_RRC_ConnectionSetup_Def();
    
          // UE shall establish SRB1 acc. to 36.331 cl. 5.3.3.4 and 5.3.9
          // UE enters RRC_CONNECTED state
          //@siclog "Step 32a3" siclog@
          // receive RRC Connection Setup Complete with piggy-backed NAS message on SRB1 (DCCH):
          v_SelectedPlmnIndex := f_NBIOT_CellInfo_GetSelectedPlmnIndex(nbiot_Cell1);
          f_NBIOT_RRCConnectionSetupComplete_Def(nbiot_Cell1,
                                                               cr_508_RRCConnectionSetupComplete_NB(tsc_RRC_TI_Def, v_SelectedPlmnIndex),
                                                               cr_NAS_IndicationWithPiggybacking (tsc_SHT_IntegrityProtected_PartiallyCiphered, cr_CONTROL_PLANE_SERVICE_REQUEST(?), cr_ESM_DATA_TRANSPORT(?, cr_UserDataContainer)));
            
          SRB.send(cas_NB_SRB_NasPdu_REQ(nbiot_Cell1,
                                               tsc_SRB1bis,
                                               cs_TimingInfo_Now,
                                               cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,
                                                              cs_SERVICE_ACCEPT)));
                
          f_NBIOT_UE_Detach_SwitchOffUe(nbiot_Cell1, CONTROL_PLANE, RRC_CONNECTED);
          
        }
      [] t_T5.timeout
        {
            //@siclog "Step 32b1" siclog@
            f_NBIOT_Postamble(nbiot_Cell1, CONTROL_PLANE, N1_IDLE);
            
        }
    }
    
    f_NBIOT_TestBody_Set(false);
    
    
  }
  
  

}



After Change:
	function f_TC_22_4_30_NBIOT() runs on NBIOT_PTC
  {// NB-IoT / NTN / Ephemeris information update / T317 Expiry / T318 Expiry @sic R5-235429, R5s230708 Ch2.1 sic@
    var NAS_MSG_Indication_Type v_NAS_Ind;
    var O10_Type v_UserData;
    var SystemInformationBlockType2_NB_r13 v_SIB2;
    var DRB_IdentityList_Type v_DrbIdList;
    var integer v_SelectedPlmnIndex;
    var integer v_NTN_delay := 0;      
    var float v_T317_Step24B := 0.0;   

      
    timer t_T317 := 60.0; //Corresponds to the value (60sec) of ul_SyncValidityDuration_r17 in SIB31-NB acc. Table 22.4.30.3.3-2
    timer t_T318 := 8.0;  //Corresponds to the value (8000ms) of t318-r17 acc 36.508 Table 4.6.3-14
    timer t_T317_Step24B;       
    timer t_T301 := 2.5;
    timer t_T5 := 30.0;
   

    //System information combination 8 as defined in TS 36.508 [18] clause 8.1.4.3.1.1 is used.
    f_NBIOT_Init(c8, CONTROL_PLANE);
      
    //Table 22.4.30.3.3-2, Value of T317=60s
    f_NBIOT_CellInfo_ModifySIB31_UL_SyncValidityDuration_r17(nbiot_Cell1,s60);
      
    //Table 22.4.30.3.3-4  @sic R5s230708 Ch2.1.2 R5-237383 sic@
    v_SIB2 := f_NBIOT_CellInfo_GetSIB2(nbiot_Cell1);
    v_SIB2.ue_TimersAndConstants_r13.t311_r13 := ms10000;
    v_SIB2.ue_TimersAndConstants_r13.t301_r13 := ms2500;
    v_SIB2.cp_Reestablishment_r14 := true_;
    f_NBIOT_CellInfo_SetSIB2 (nbiot_Cell1, v_SIB2);
    v_NTN_delay := f_NBIOT_NTN_Get_T_NTNoffset(nbiot_Cell1);
    v_T317_Step24B := 60.0 - int2float(v_NTN_delay)/1000.0;
    
    //Create and configure all cells
    f_NBIOT_CellConfig_Def(nbiot_Cell1, CONTROL_PLANE);
      
    f_UT_SwitchOnUE(UT, false);
    //The UE is in Automatic PLMN selection mode
    f_UT_AutomaticPLMN_Select_WithSwitchOnOff (UT);
      
    //The preconfigured UE location is defined in TS 36.508 [18] Clause 4.13.
    f_UT_ConfigureNTNPosition(UT);
    
    //Preamble: The UE is in State 1-NB switched off according to TS 36.508 [18].
    f_UT_PowerOffUE (UT);
      
    f_NBIOT_SetCellPower(nbiot_Cell1, tsc_ServingCellRS_EPRE);
    
    f_NBIOT_TestBody_Set(true);
    
    //@siclog "Step 1" siclog@
    //Power on UE.
    f_UT_SwitchOnUE(UT, false);
      
    //@siclog "Step 2" siclog@
    //UE acquires SystemInformationBlockType31-NB and start T317 = 60 seconds.
    
    //@siclog "Steps 3-18b1" siclog@
    //Steps 1 to 16b1 of the Attach, connected mode and test mode activated procedure described in TS 36.508 [18] subclause 8.1.5.2A.3 with TEST LOOP MODE A are performed on Ncell 1.
    // Acc to 36.508 Table 8.1.5.2A.3-1: NB-IoT UE Attach, connected mode and test mode activated procedure (state 2-NB to state 2A-NB)
    //Note: After receiving the message in step 4, SS start a timer T1 with value = T317.
    v_NAS_Ind := f_NBIOT_InitialRegistration_Step2_4(nbiot_Cell1, CONTROL_PLANE, PREAMBLE); //@sic R5230708 Ch2.1.3 sic@
    t_T317.start;    //T1 started
    f_NBIOT_InitialRegistration_Step5_14(nbiot_Cell1, CONTROL_PLANE, STATE2A_NB_TESTLOOP_ModeG, v_NAS_Ind);
         
    //@siclog "Steps 18A-18B" siclog@
    //Steps 1 to 2 of the UE Test Loopback Activated procedure described in TS 36.508 [18] subclause 8.1.5.2B.3 are performed on Ncell 1.
    f_NBIOT_CloseUE_TestLoopModeGH (nbiot_Cell1, tsc_UE_TestLoopMode_TypeG, tsc_TestLoopModeGH_UplinkMode_NAS); //@sic R5230708 Ch2.1.4 sic@
      
    //@siclog "Step 19" siclog@
    //SS stops broadcasting SystemInformationBlockType31-NB
    f_NBIOT_ModifySysinfoCombinationTo_C1(nbiot_Cell1, true, false);  //now only SIB2 will be broadcasted with cellBarred and  DottachWithoutPDN = false for CONTROL_PLANE
    f_NBIOT_SS_ConfigureSysinfo(nbiot_Cell1, false);
     
    //@siclog "Step 19A" siclog@
    //SS stops periodic grant allocation
    f_NBIOT_ULGrantTransmission(nbiot_Cell1, cs_NB_UL_GrantScheduling_Stop);
      
    //@siclog "Step 20" siclog@
    //The SS transmits one IP packet embedded in an ESM DATA TRANSPORT and DLInformationTransfer-NB. (Note 1)
    v_UserData := f_GetN_OctetsFromPRBS(0, 10);
    SRB.send(cas_NB_SRB_NasPdu_REQ(nbiot_Cell1,
                                     tsc_SRB1bis,
                                     cs_TimingInfo_Now,
                                     cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,
                                                    cs_ESM_DATA_TRANSPORT (tsc_EpsDefaultBearerId, tsc_PTI_Unassigned, cs_UserDataContainer(v_UserData))))); //@sic R5s230708 Ch.2.1.6 sic@
      
      
    //@siclog "Steps 21-22" siclog@
    //Void
      
    //@siclog "Step 23" siclog@
    //Wait until expiry of T1.
    //Note: UE starts timer T318. SS start a timer T2 with value = T318.
    t_T317.timeout;  //T1 expires
    t_T318.start;    //T2 started
     
    //@siclog "Step 24" siclog@
    //The SS starts the UL periodic grant transmission, allowing the UE to return the IP packet as received in step 20
    f_NBIOT_ULGrantTransmission(nbiot_Cell1, cs_NB_UL_GrantScheduling_Start(cs_NB_ContinuousGrant)); //@sic R5s230708 Ch.2.1.7 sic@
            
    //@siclog "Step 24A" siclog@
    //SS resumes broadcasting SystemInformationBlockType31-NB.
    f_NBIOT_ModifySysinfoCombinationFromC1_To_C8(nbiot_Cell1, s60, false);  //now SIB31-NB will be broadcasted again with T317=60s and DoAttachWithoutPDN=false for CONTROL_PLANE
    f_NBIOT_SS_ConfigureSysinfo(nbiot_Cell1, false);
    //Note: Timer T318 shall be stopped and T317 shall be started in UE side. SS starts a timer T3 with value = T317.
    t_T318.stop;     //T2 stopped
      
    //@siclog "Step 24B" siclog@
    //Check: Does the UE transmit one IP packet embedded in an ESM DATA TRANSPORT and ULInformationTransfer-NBcorresponding to grant in step 24?
    SRB.receive(car_NB_SRB_NasPdu_IND(nbiot_Cell1,
                                      tsc_SRB1bis,
                                      cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,
                                      cr_ESM_DATA_TRANSPORT(?, cr_UserDataContainer))));
    f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 24B");

    t_T317_Step24B.start(v_T317_Step24B);

    //@siclog "Step 24C" siclog@
    //SS stops broadcasting SystemInformationBlockType31-NB.
    f_NBIOT_ModifySysinfoCombinationTo_C1(nbiot_Cell1,true,false);  //now only SIB2 will be broadcasted with cellBarred and  DottachWithoutPDN = false for CONTROL_PLANE
    f_NBIOT_SS_ConfigureSysinfo(nbiot_Cell1, false);
      
    //@siclog "Step 24CA" siclog@
    //SS stops periodic grant allocation
    f_NBIOT_ULGrantTransmission(nbiot_Cell1, cs_NB_UL_GrantScheduling_Stop);
    
    //@siclog "Step 24D" siclog@
    //The SS transmits one IP packet embedded in an ESM DATA TRANSPORT and DLInformationTransfer-NB (Note 1).
    v_UserData := f_GetN_OctetsFromPRBS(10, 10);
    SRB.send(cas_NB_SRB_NasPdu_REQ(nbiot_Cell1,
                                     tsc_SRB1bis,
                                     cs_TimingInfo_Now,
                                     cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,
                                                    cs_ESM_DATA_TRANSPORT (tsc_EpsDefaultBearerId, tsc_PTI_Unassigned, cs_UserDataContainer(v_UserData))))); //@sic R5s230708 Ch.2.1.6 sic@
      
      
    //@siclog "Step 24E" siclog@
    //Wait until expiry of T3.
    //Note: UE starts timer T318. SS starts a timer T4 with value = T318
    t_T317_Step24B.timeout;   //T3 expired 
    t_T318.start;               
    
    //@siclog "Step 24F" siclog@
    //SS starts periodic UL grant transmission, allowing the UE to return the IP packet as received in step 24D.
    f_NBIOT_ULGrantTransmission(nbiot_Cell1, cs_NB_UL_GrantScheduling_Start(cs_NB_ContinuousGrant)); //@sic R5s230708 Ch.2.1.7 sic@
     
      
    //@siclog "Step 25" siclog@
    //Check: Does the UE transmit one IP packet embedded in an ESM DATA TRANSPORT and ULInformationTransfer-NB corresponding to grant in step 24F?
    alt{
      [] SRB.receive(car_NB_SRB_NasPdu_IND(nbiot_Cell1,
                                      tsc_SRB1bis,
                                      cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,
                                      cr_ESM_DATA_TRANSPORT(?, cr_UserDataContainer))))
    
        {
            f_NBIOT_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 25");
        }
    
      [] t_T318.timeout  //T4 expires @sic R5s230708, R5-237383 sic@   
        {
          //@siclog "Step 26" siclog@
          //Wait until expiry of T4.
        }
    }
     
    //@siclog "Step 26A" siclog@
    //SS resumes broadcasting SystemInformationBlockType31-NB.
    f_NBIOT_ModifySysinfoCombinationFromC1_To_C8(nbiot_Cell1, s60, false);  //now SIB31-NB will be broadcasted again with T317=60s and DoAttachWithoutPDN=false for CONTROL_PLANE
    f_NBIOT_SS_ConfigureSysinfo(nbiot_Cell1, false);
    
    //EXCEPTION: Steps 27-30 describe behaviour that depends on whether the UE supports RRC connection re-establishment.
    if(pc_RRC_re_establishment_CP_CIoT)
    {//@sic R5s230708 Ch2.1.9 R5-237383 sic@
        // Locally release
        v_DrbIdList := f_NBIOT_SS_ActiveDRBs(nbiot_Cell1, CONTROL_PLANE);
        f_NBIOT_SS_SRBs_DRBs_Release(nbiot_Cell1, CONTROL_PLANE, -, v_DrbIdList);
        f_NBIOT_SS_SRBs_DRBs_Config(nbiot_Cell1, CONTROL_PLANE, -, omit);
        f_NBIOT_SS_ConfigRachProcedureWithNoMsg4(nbiot_Cell1);

        //@siclog "Step 27" siclog@
        //Check: Does the UE transmit an RRCConnectionReestablishmentRequest-NB message on SRB0?
        f_NBIOT_RRC_RRCConnectionReestablishmentRequest_r14(nbiot_Cell1);
        f_NBIOT_PreliminaryPass(__FILE__, __LINE__, " Step 27");
        
        //@siclog "Step 28" siclog@
        //The SS does not respond for 2.5s (T301) and the SS shall handle any re-transmitted RRCConnectionReestablishmentRequest-NB message on Ncell 2
        t_T301.start;
        alt {
          [] SRB.receive(car_NB_SRB0_RrcPdu_IND(nbiot_Cell1, cr_RRCConnectionReestablishmentRequest_NB_r14(?, otherFailure)))
            {
              repeat;
            }
          [] t_T301.timeout {}
        }
       
        //@siclog "Step 29" siclog@
        //Void
        
        //@siclog "Step 30" siclog@
        //Void
        
        f_NBIOT_SS_ConfigRachProcedureMsg4(nbiot_Cell1, f_NBIOT_508_RRCConnectionSetup());
    }
    else
    {
        // Locally release the connection as UE will go to idle mode
        v_DrbIdList := f_NBIOT_SS_ActiveDRBs(nbiot_Cell1, CONTROL_PLANE);
        f_NBIOT_SS_SRBs_DRBs_Release(nbiot_Cell1, CONTROL_PLANE, -, v_DrbIdList);
        f_NBIOT_SS_SRBs_DRBs_Config(nbiot_Cell1, CONTROL_PLANE, -, omit);
    }
      
    
    //@siclog "Step 31" siclog@
    t_T5.start; //@sic R5s230708, R5-237383 sic@
    alt {//@sic R5s230708, R5-237383 sic@
      //@siclog "Step 32a1" siclog@
      [] SRB.receive(car_NB_SRB0_RrcPdu_IND(nbiot_Cell1, cr_RRCConnectionRequest_NB(?, ?)))
        {
          //@siclog "Step 32a2" siclog@
          t_T5.stop;

          // switch on UL grant assignments
          f_NBIOT_ULGrantTransmission(nbiot_Cell1, cs_NB_UL_GrantScheduling_Start(cs_NB_ContinuousGrant));
          f_NBIOT_RRC_ConnectionSetup_Def();
    
          // UE shall establish SRB1 acc. to 36.331 cl. 5.3.3.4 and 5.3.9
          // UE enters RRC_CONNECTED state
          //@siclog "Step 32a3" siclog@
          // receive RRC Connection Setup Complete with piggy-backed NAS message on SRB1 (DCCH):
          v_SelectedPlmnIndex := f_NBIOT_CellInfo_GetSelectedPlmnIndex(nbiot_Cell1);
          f_NBIOT_RRCConnectionSetupComplete_Def(nbiot_Cell1,
                                                               cr_508_RRCConnectionSetupComplete_NB(tsc_RRC_TI_Def, v_SelectedPlmnIndex),
                                                               cr_NAS_IndicationWithPiggybacking (tsc_SHT_IntegrityProtected_PartiallyCiphered, cr_CONTROL_PLANE_SERVICE_REQUEST(?), cr_ESM_DATA_TRANSPORT(?, cr_UserDataContainer)));
            
          SRB.send(cas_NB_SRB_NasPdu_REQ(nbiot_Cell1,
                                               tsc_SRB1bis,
                                               cs_TimingInfo_Now,
                                               cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,
                                                              cs_SERVICE_ACCEPT)));
                
          f_NBIOT_UE_Detach_SwitchOffUe(nbiot_Cell1, CONTROL_PLANE, RRC_CONNECTED);
          
        }
      [] t_T5.timeout
        {
            //@siclog "Step 32b1" siclog@
            f_NBIOT_Postamble(nbiot_Cell1, CONTROL_PLANE, N1_IDLE);
            
        }
    }
    
    f_NBIOT_TestBody_Set(false);

